
;
;  BOOT BLOCK.
;
;  AUTHOR: PAUL HARDY.  THIS CODE IS IN THE PUBLIC DOMAIN.
;
;  THIS CODE IS IN TRACK 0 SECTOR 0 OF THE RL02, FOR STAND-ALONE BOOT.
;
;  THIS BOOT BLOCK READS SCREENS 40-47 OF THE ’FORTH.DAT’ FILE.
;  WHEN THE LOADER EXECUTES IT READS IN THE BINARY IMAGE (STARTING AT
;  1000 OCTAL) AND BRANCHES TO ADDRESS 1000 TO START EXECUTION OF THE
;  LOADED PROGRAM.
;
        .ASECT
.=0
ORIGIN: NOP
        BR      BOOTRL
.=ORIGIN+4
        .WORD   ERROR           ; TIME OUT / BUS ERROR VECTOR
        .WORD   343             ; FLAG (WHICH HALT)
.=ORIGIN+10
        .WORD   ERROR           ; RESERVED INSTRUCTION VECTOR
        .WORD   344             ; FLAG (WHICH HALT)
        .WORD   DEBUG           ; EXECUTE WAIT INSTRUCTION
        .WORD   340             ; SET PSW = PRIORITY 7
.=ORIGIN+20
        .WORD   WAIT            ; IOT TRAP - USED FOR WAIT SUBROUTINE
        .WORD   340             ; SET PSW = PRIORITY 7
        .WORD   ERROR           ; POWER FAILURE INTERRUPT VECTOR
        .WORD   345             ; FLAG (WHICH HALT)
.=ORIGIN+30
DEBUG:  WAIT                    ; EMT TRAP AREA, USED AS DEBUG VECTOR
        RTI
.=ORIGIN+34
        .WORD   ERROR           ; If TRAP, halt with PSW=000000
ERROR:  HALT
.=ORIGIN+60
        .WORD   ERROR           ; HALT IF TERMINAL INPUT INTERRUPT
        .WORD   341             ; FLAG (WHICH HALT)
        .WORD   ERROR           ; HALT IF TERMINAL OUTPUT INTERRUPT
        .WORD   342             ; FLAG (WHICH HALT)
.=ORIGIN+100
        .WORD   RETURN          ; IF KW11L CLOCK INTERRUPT, JUST RTI
        .WORD   340
        .WORD   RETURN          ; IF KW11P CLOCK INTERRUPT, JUST RTI
        .WORD   340
;
; Declarations for terminal I/O
;
RCSR=177560     ; Terminal receive control and status
RBUF=177562     ; Terminal receive buffer
XCSR=177564     ; Terminal transmit control and status
XBUF=177566     ; Terminal transmit buffer
;
; Declarations for RL02 Disk Controller
;
RLCS=174400
RLBA=174402
RLDA=174404
RLMP=174406

BOOTRL:
        MOV     #RLCS,R0
        ;
        ; GET STATUS
        ;
        MOV     #13,4(R0)       ; RLDA SETUP FOR GET STATUS

Page 1



00000000011111111112222222222333333333344444444445555555555666666666677777777778
12345678901234567890123456789012345678901234567890123456789012345678901234567890
        MOV     #4,(R0)         ; RLCS GET STATUS
        TSTB    (R0)            ; RLCS TEST FOR READY STATE
        BPL     .-2
        MOV     6(R0),R1        ; RLMP STATUS COPY TO R1
        ;
        ; SEEK CYLINDER 2, UPPER HEAD, FROM CYLINDER 0
        ;
        MOV     #000405,4(R0)
        MOV     #000006,(R0)    ; SEEK COMMAND
        TSTB    (R0)            ; RLCS TEST FOR READY STATE
        BPL     .-2
        ;
        ; READ DATA
        ;
        MOV     #000400,2(R0)   ; RLBA START LOAD AT ADDR 000400 OCTAL
        MOV     #000431,4(R0)   ; RLDA CYLINDER 2, HEAD 0, SECTOR 31 (OCTAL)
        MOV     #-3600,6(R0)    ; RLMP GET 15. (DECIMAL) SECTORS = 3.75 KB
                                ;      (DON’T LOAD BOOT SECTOR)
        MOV     #14,(R0)        ; RLCS READ DISK WITH HEADER CHECK
        TSTB    (R0)            ; RLCS TEST FOR READY STATE
        BPL     .-2
        MOV     (R0),R3         ; RLCS READ STATUS COPY TO R3
        ;
        ; SEEK CYLINDER 2, LOWER HEAD, FROM CYLINDER 0
        ;
        MOV     #000020,4(R0)
        MOV     #000006,(R0)    ; SEEK COMMAND
        TSTB    (R0)            ; RLCS TEST FOR READY STATE
        BPL     .-2
        ;
        ; READ DATA
        ;
        MOV     #010000,2(R0)   ; RLBA START LOAD AT ADDR 010000 OCTAL
        MOV     #000500,4(R0)   ; RLDA CYLINDER 2, HEAD 1, SECTOR 0
        MOV     #-4000,6(R0)    ; RLMP GET 16. (DECIMAL) SECTORS = 4 KB
        MOV     #14,(R0)        ; RLCS READ DISK WITH HEADER CHECK
        TSTB    (R0)            ; RLCS TEST FOR READY STATE
        BPL     .-2
        MOV     (R0),R5         ; RLCS READ STATUS COPY TO R3
        ;
        ; BEGIN FORTH EXECUTION
        ;
        JMP     1000
        .END

